CHAPTER 5

NUMERICAL SOLUTIONS TO KINEMATIC FLOW

NUMERICAL METHODS

The kinematic method has a number of advantages over other methods:

1. The mathematics are simpler than those of the comprehensive hydrodynamic equations, although hydrodynamic forces are omitted.
2. It is relatively simple to visualize the flow process described by the kinematic equations.
3. The gradual increase in water depth over a catchment during a storm can be simulated. This allowance for dynamic effects is not possible with isochronal techniques. The latter techniques use only the friction equation for steady state flow with no allowance for continuity.
4. The effect of storm intensity influences the concentration time of a catchment as it should. This is not the case with isochronal methods.
5. The equations are amenable to analytical solution in many cases.
6. Numerical solutions are feasible and simple for non-rectangular catchments, varying topography, spatial and temporal variation of storms and losses and combinations of overland and conduit flow.

Various workers (eg. Overton and Meadows, 1976) have employed the kinematic equations for catchment models.

EFFECT OF STORM DISTRIBUTION ON RUNOFF

Many hydrological studies are made on the assumption of a rectangular catchment and uniform storm distribution in space and time. Numerical simulation models such as SWMM (the overland flow components) and analytical models such as those of Wooding (1965) are based on rectangular basins. The effect of an uneven and non-planar basin can be to increase the intensity of runoff for a storm of any particular return period. A basin with its centre of gravity close to the mouth can result in a more severe runoff than a long rectangular basin or one with the centre of gravity further up the basin. Similarly a storm with a focus close to the mouth of the basin will result in a more intense runoff than a storm which is uniformly spread over the catchment.

An allowance for non-uniform storm distribution can be made with the tangent method of design. That technique, however, is based on uniform flow down the basin and the time of concentration is therefore inaccurately predicted. The true dynamic concentration of the storm in the basin must be predicted from the equations of motion. In many cases these can be approximated by the kinematic equations. These equations are capable of analytical solution in many cases of uniform shaped plane basins and uniform storms. For irregular shaped basins and hyetographs, it is necessary to solve these or the rigorous hydrodynamic equations numerically.

It is the purpose of this section to demonstrate with the aid of the kinematic equations of flow how catchment shape and hyetograph shape affect the discharge hydrograph shape and effective time of concentration. A planar quadrilateral shaped basin is assumed and lateral flow time (perpendicular to the general direction of flow) is assumed negligible. In one case the basin is assumed to be rectangular with the slope in one direction parallel to two opposite sides. The excess rain is assumed to increase from zero to a maximum and then decrease to zero again over a defined time. The resulting hyetograph is triangular. This distribution in one extreme case could also account for a time-varying infiltration rate. In another case, the excess rain is assumed to vary spatially in a triangular fashion from zero at the top of the basin to a maximum along the basin. Thus the basin may be rectangular with the storm varying in intensity down the length of the basin (Fig. 5.4) or the storm could be of uniform intensity while the basin width varies down the length (Fig. 5.5).

GENERAL EQUATIONS

Start with the basic kinematic equations:

Continuity
$$\frac{\partial y}{\partial t} + \frac{\partial q}{\partial x} = i_e \tag{5.1}$$

Dynamic equilibrium
$$q = z\,y^n \tag{5.2}$$

or
$$q = \frac{7.7\sqrt{Sg}}{k^{1/6}}\,y^{5/3} \tag{5.3}$$

where $z = 7.7\sqrt{Sg}\,/k^{1/6}$ and $n = 5/3$ employing the Manning-Strickler equation.

Equation 5.2 is the more general equation for uniform flow but the constants are dimensionally dependent.

In these equations, z is a factor, n is a coefficient, x is distance in the flow direction, t is time, y is flow depth, g is gravitational acceleration, S is the bed slope, equal to the friction energy loss gradient, q is the discharge rate per unit width, i_e is the excess rainfall rate per unit area after subtracting infiltration and other losses, and k is a measure of surface roughness.

Substituting for y from (5.3) or (5.2) in (5.1) we get

$$\frac{1}{nz^{1/n}q^{1-1/n}} \frac{\partial q}{\partial t} + \frac{\partial q}{\partial x} = i_e \tag{5.4}$$

or $$\frac{0.6}{q^{0.4}(7.7\sqrt{Sg}/k^{1/6})^{0.6}} \frac{\partial q}{\partial t} + \frac{\partial q}{\partial x} = i_e \tag{5.5}$$

The equation may be rendered dimensionless by substituting

$$P = q/Li_{ea} \tag{5.6}$$
$$I = i_e/i_{ea} \tag{5.7}$$
$$X = x/L \tag{5.8}$$
$$T = (zi_{ea}^{\ n-1}/L)^{1/n}t = \frac{7.7^{0.6}(Sg)^{0.3}i_{ea}^{\ 0.4}t}{L^{0.6}k^{0.1}} \tag{5.9}$$

where i_{ea} *is the time and space averaged excess rainfall rate over the* catchment and L is the length of catchment in the direction of flow.
Now (5.5) becomes $$\frac{\partial P}{\partial T} = n(pP)^{1-1/n}(I - \frac{\partial P}{\partial X}) \tag{5.10}$$

$$= 2.2P^{0.4}(I - \frac{\partial P}{\partial X}) \tag{5.11}$$

where p is the ratio of peak to average excess rain intensity i_{ep}/i_{ea} which is 2 for a triangular distribution.
It may also be proved that

$$T = t/t_c \tag{5.12}$$

where t_c is the time to equilibrium or concentration time of a rectangular plane catchment subject to uniform (in time and space) excess rain. For that case analytical solutions to the kinematic equations are feasible. Thus the rising limb of the hydrograph at the end of a catchment is given by

$$q = i_eL(t/t_c)^{5/3} \tag{5.13}$$

where $$t_c = (L/zi_{ea}^{\ n-1})^{1/n} \tag{5.14}$$

$$= \frac{L^{0.6}k^{0.1}}{7.7^{0.6}(Sg)^{0.3}i_{ea}^{\ 0.4}} \tag{5.15}$$

hence $$P = T^{5/3} \tag{5.16}$$

The falling limb of the hydrograph, beyond $t = t_c$ is given by the implicit equation

$$L = zy_L^{n-1}\left[y_L/i_e + n(t-t_d)/y_L\right] \tag{5.17}$$

which may be rewritten for the case of rain duration t_d equal to time of concentration t_c as

$$T = 1 + (1-P)/(nP^{1-1/n}) \tag{5.18}$$

$$= 1 + (1-P)/(\tfrac{5}{3} P^{0.4}) \tag{5.19}$$

The rising and falling limbs may be plotted from (5.16) and (5.19) as in Fig. 5.3. For other cases, where rainfall is not uniform in time and space, numerical solutions are necessary. For this purpose (5.9) was solved for specific cases.

SOLUTIONS FOR NON UNIFORM AND UNSTEADY STORM INPUT

Equation (5.9) was solved iteratively using a backward difference explicit finite difference solution technique. Acceptable accuracy was obtained with $\Delta X = 0.05$ and $\Delta T = \Delta X$ gave a stable solution. The finite difference form of the equation became

$$P(X_i,T_j) = P(X_i,T_{j-1})+2.2\ \Delta T\ P(X_i,T_{j-1})^{0.4}\{I-[P(X_i,T_{j-1}) - P(X_{i-1},T_{i-1})]/\Delta X\} \tag{5.20}$$

At T = 0 this explicit form of (5.9) would yield zero increment in P(x) so a centred explicit - implicit numerical form was employed i.e.

$$P(X_i,T_2) = 0 + 2.2\ I\ \Delta T[P(X_i,T_1) + P(X_i,T_2)]^{0.4}/2 \tag{5.21}$$

$$\therefore\ P(X_i,T_2) = (2.2\ I\Delta T/2)^{5/3} \tag{5.22}$$

The scheme was used to study two particular cases:



Fig. 5.1 Excess rain variation with time

I. Time varying rain intensity

A rectangular plane basin sloping in the longitudinal direction was analyzed for different cases of unsteady rainfall input. Excess rainfall was assumed to be spread uniformly over the basin, but it was permitted to vary in time. The storm duration was set equal to the concentration time of a rectangular basin subject to uniform excess rain rate equal to the mean for the non-uniform case. A triangular hyetograph was assumed with the peak varying successively in time for different cases from the start of the storm to the end of the storm, i.e. $F = t_p/t_d = 0$ to 1 where t_d is the storm duration and t_p is the time to peak of the storm (see Fig. 5.1).

The case could be applied to the excess rain after subtracting losses, infiltration etc. Alternatively a decaying rate of infiltration could be allowed for. A rectangular storm hyetograph with a straight line decrease in losses may in some cases be approximated by a triangular excess rain hyetograph with the peak at the end of the storm (Fig.5.2).



Fig. 5.2 Rectangular hyetograph with decaying losses resulting in triangular excess rain distribution with time

The resulting runoff hydrographs at the mouth of the basin are depicted in Fig. 5.3. Thus it will be observed that as F_t (the relative time to storm peak) varies from 0 to 1, so the effective concentration time of the basin increases, i.e. the time to peak of the runoff hydrograph increases. The peak runoff also increases the later the storm peak although it is always less than the peak excess rain rate which is $2i_{ea}L$ per unit width for a triangular storm distribution. In fact the worst storm distribution occurs with a storm peak at the end of the storm, and for this case the peak runoff is $1.4\ i_{ea}L$ per unit width.

For comparison the runoff hydrograph for a rectangular storm is indicated in dashed lines in Fig. 5.3. The rising limbs of hydrographs for storm durations exceeding the nominal concentration time of the basin are also indicated. Here the storm intensity is assumed to increase linearly with time, reaching a maximum at the end of the storm ($t_p = t_d$).



Fig. 5.3: Runoff hydrographs for triangular storm time distribution. $F_t = t_p/t_c$ where t_p = time to peak of excess rain and storm duration $t_d = t_c$ = concentration time for rectangular storm input except for F>1

If the precipitation rate was constant over a duration equal to the concentration time of the catchment, t_c, then the peak runoff would occur at the end of the storm and would be equal to $i_{eu}L$ per unit width of catchment, where i_{eu} is the uniform excess rainfall rate. This case is generally accepted as the worst storm for any particular return period and the duration of the design storm is thus selected. In the case of an unsteady (time varying) storm it is difficult to solve directly for the storm duration which will result in the maximum peak runoff. According to Fig. 5.3 for the case of a storm peaking at the end (F_t = 1) the peak runoff occurs at $t = t_c$ so a storm duration equal to t_c will result in maximum peak runoff, whereas for a storm peaking at the beginning (F_t = 0) the peak occurs before $t = t_c$ so a shorter storm will probably result in the maximum peak runoff. Also a storm with a longer duration than t_c may result in the maximum peak runoff if the storm peak occurs at the end.

It will be apparent from Fig. 5.3 that the worst storm with duration t_c is that which peaks at the end (F_t = 1.0). A time decaying infiltration loss can have the same effect on a uniform storm. A rectangular hyetograph could be transformed to a triangular one with the maximum excess rain at the end. In this case the peak runoff would be 40% higher than that for a storm with a constant loss rate.

II. Space varying rain input

Consider the case of a rectangular plane basin with a steady storm intensity varying down the length of the basin. Storm intensity is assumed constant with time and the excess rain intensity is triangular with a peak somewhere along the basin, defined by $F_x = x_p/L$ between zero and one (Fig. 5.4). The rising limb of the resulting discharge hydrograph is depicted in Fig. 5.6.



Fig. 5.4 Excess rain variation along basin

It will be observed that in all cases the hydrograph peaks near the concentration time of a uniform storm hydrograph (the dashed line in Fig. 5.6 is for a uniform storm in time and space). The hydrograph rises much faster in the case of a storm whose peak is at the mouth of the basin ($F_x = 1$) and much slower in the case of the storm with a peak at the top end of the basin ($F_x = 0$). A storm with a duration less than the nominal concentration time of the basin may therefore result in a higher peak runoff if its centre of gravity is near the mouth of the basin.



Fig. 5.5 Varying basin width

The same chart applied to the case of a basin whose width varies from zero to a miximum somewhere along the basin and then back to zero at the end (Fig. 5.5). Provided lateral flow time can be neglected, the effective excess rain input at any point x along the collecting channel of width w is $i_e' = i_{eu}W/w$ where i_{eu} is the uniform excess rainfall rate and W is the basin width. The mean excess rainfall is $i_{ea}' = i_{eu}W_{max}/2w$. The values of i_e' and i_{ea}' are used in place of i_e and i_{ea} respectively in Fig. 5.5 to yield the discharge q per unit width of channel. If lateral flow time is significant it is necessary to correct for this. An approximation is to move the true storm peak upbasin by the lateral flow distance.

It is evident that a basin which is wider at the mouth than upstream will result in a hydrograph which rises more rapidly initially than for a rectangular basin. Thus a storm duration less than the concentration time of the basin may result in the maximum peak runoff. It would be necessary to obtain the storm duration resulting in maximum peak runoff by trial and error using Fig . 5.6 for calculating the peak runoff intensity corresponding to the selected storm duration.



Fig. 5.6 Rising discharge hydrograph for triangular runoff contribution along watercourse. $F_x = x_p/L$ where x_p = distance to peak input and L = catchment length

Multiple variations

Fig. 5.7 applies to a storm which increases linearly in intensity towards the mouth of a rectangular catchment and also increases linearly with time. For this figure i_{ea} is taken as the excess rainfall rate occuring half way down the catchment at a time equal to half the concentration time for a uniform storm. It will be observed that the peak runoff for a storm with duration t_c would be 1.65 times the mean excess rainfall rate multiplied by basin area. This is even higher than that resulting from a time-increasing storm with even distribution down the basin (Fig. 5.3) as could be expected.



$$T = \frac{7.7^{0.6}(Sg)^{0.3}\, i_{ea}^{0.4}\, t}{L^{0.6}\, k^{0.1}} = \frac{t}{t_c}$$

Fig. 5.7 Rising discharge hydrograph for rainfall increasing uniformly towards mouth of rectangular basin and uniformly in time. i_{ea} occurs at X = 0.5 and T = 0.5.



Fig. 5.8 Storm travelling along basin

Fig. 5.9 depicts the hydrographs from a rectangular basin with a travelling storm. For two cases the storm distribution down the basin is triangular. The peak of the triangle is assumed to travel at a speed L/t_c up or down the basin (see Fig. 5.8) but the storm is confined to the basin for its duration. The storm duration is taken as t_c. The peak runoff intensity for a storm travelling down the basin is more severe than for a storm travelling up the basin.

Also depicted on Fig. 5.9 are the discharge hydrographs for a rectangular storm travelling up or down the basin. The excess rain is uniform over the area of precipitation, and the storm is across the entire width of the basin, but is of limited longitudinal extent. The length of storm path is assumed equal to the length of the basin, and the storm front travels up or down the basin, starting at one end and continuing along and beyond the basin (Fig. 5.10).

SUMMARY OF EFFECTS OF STORM DISTRIBUTION

The effect of non uniform storm distribution, whether in space or time, is generally to increase peak runoff from a basin. Simple triangular storm distributions were analyzed by numerical solution of the kinematic flow equations to illustrate the effect and to produce generalized design charts.

A storm which peaks near the end of its duration can cause a runoff intensity 40 percent higher than a uniform storm of the same average intensity. The same effect manifests with a decaying infiltration rate. Storms which are more concentrated near the mouth of the basin than further upstream may result in a higher runoff than a uniformly spread



Fig. 5.9 Discharge hydrographs due to travelling storm of duration t_c



Fig. 5.10 Travelling rectangular storm

storm. Similarly basins which have the centre of gravity of the area near the mouth result in higher peak runoff than rectangular basins.

A storm travelling down a catchment will result in higher peak runoff than the same storm travelling upstream.

The assumption of uniform storm intensity and distribution can only yield average figures for any selected return period of rainfall. The number of variables contributing to the intensification of runoff imply the probability of exceedance of a particular runoff rate may be greater than is indicated by a statistical analysis of isolated rain gauge records.

TWO-DIMENSIONAL MODELS

The assumption of a one-dimensional flow off a rectangular plane catchment is often inaccurate. Many catchments vary topographically in two dimensions. Hills and valleys cause runoff to flow in varying directions. Flow will at all times be perpendicular to the contour lines under the assumption of kinematic flow. In addition to the flow path due to the lateral flow, lateral slopes may also result in flow concentrations in valleys with resulting effect on concentration time. Thus for large catchments a two-dimensional analysis is desirable.

The kinematic equations may readily be generalized for the two-dimensional case.

The continuity equation becomes

$$\frac{\partial y}{\partial t} + \frac{\partial q_x}{\partial x} + \frac{\partial q_y}{\partial y} = i_e \tag{5.23}$$



Fig. 5.11 Topography in 3-dimensions

There are two components of flow in the x and y directions, and consequently two discharge-depth equations:

$$q_x = \frac{1}{q_t} (\alpha_x y^m)^2 \qquad (5.24)$$

$$q_y = \frac{1}{q_t} (\alpha_z y^m)^2 \qquad (5.25)$$

where $q_t = \sqrt{q_x^2 + q_z^2}$, $\alpha_x = 7.7\sqrt{gS_{ox}}/k^{1/6}$ (similarly for α_z), and m = 5/3 in the Manning-Strickler equation. The resulting equations can be solved across an x-z grid at successive time increments using numerical methods Constantinides (1981) found that a backward central explicit difference scheme yielded satisfactory results with minimal computational time.

Two-dimensional models can readily be extended to allow for varying surface roughness losses, canalization and storm distributions in time and space. Where discontinuities are present such as in built up areas, it may be easier to use one of the available simplistic runoff models such as SWMM or ILLUDAS.

ANALYSIS OF FLOW IN PART-FULL PIPES

Simulation of flow in part-full circular conduits is more complicated than for overland flow. Nevertheless the same basic kinematic theory applies. It is convenient to adopt polar axes instead of cartesian axes for circular conduits. On this basis a computer program for analysing flow in pipe networks was prepared. The system is assumed to comprise overland flow planes connected to pipe inlets.

Overland flow

In analyzing the flow over the sub-catchments draining into each inlet, the assumption is made that the sub-catchment is rectangular with a width equal to the length of drain pipe into which the sub-catchment drains. This simplification is mainly to reduce data preparation to a minimum, and the relevant input line could readily be varied to feed in sub-catchment width. For overland flow the cross-sectional area per unit width of catchment is y. The concentration time of a plane is, adopting the Manning-Strickler equation for friction gradient,

$$t_c = \left\{ \frac{L\ k^{1/6}}{7.7\sqrt{Sg}\ i_e^{2/3}} \right\}^{3/5} \qquad (5.26)$$

Kinematic equations for part-full circular conduits

One starts with the kinematic equations in the form

$$\frac{\partial A}{\partial t} + \frac{\partial Q}{\partial x} = q \quad \text{(inflow per unit length)} \tag{5.27}$$

$$S_o = S_f \tag{5.28}$$

Using the Manning equation to predict friction gradient,

$$Q = \frac{K}{N}\frac{A^{5/3}}{P^{2/3}} S^{\frac{1}{2}} \tag{5.29}$$

and from the Strickler approximation for N,

$$N = 0.13Kk^{1/6}/\sqrt{g} \quad \text{therefore} \tag{5.30}$$

$$Q = \frac{7.7\sqrt{Sg}}{k^{1/6}}\frac{A^{5/3}}{P^{2/3}} \tag{5.31}$$

No allowance for losses at manholes is made as this is usually included in the grading of successive pipes.

The cross-sectional area of flow in a circular conduit running part full (see Fig. 5.12) is

$$A = \frac{D^2}{4}\left(\frac{\Theta}{2} - \cos\frac{\Theta}{2}\sin\frac{\Theta}{2}\right) \tag{5.32}$$

$$\text{and } P = D\frac{\Theta}{2} \tag{5.33}$$

Thus if one takes Θ as the variable, the continuity equation becomes

$$\frac{D^2}{8}\left(1 + \sin^2\frac{\Theta}{2} - \cos^2\frac{\Theta}{2}\right)\frac{\partial\Theta}{\partial t} + \frac{\partial Q}{\partial x} = q \tag{5.34}$$

In finite difference form, solving for Θ after a time interval Δt,

$$\Theta_2 = \Theta_1 + \left(q - \frac{\Delta Q}{\Delta x}\right)\frac{8\Delta t}{D^2\left(1+\sin^2\frac{\Theta}{2} - \cos^2\frac{\Theta}{2}\right)} \tag{5.35}$$

and in terms of the new Θ,

$$Q = \frac{7.7\sqrt{Sg}}{k^{1/6}}\frac{D^2}{4}\left(\frac{\Theta}{2} - \cos\frac{\Theta}{2}\sin\frac{\Theta}{2}\right)\left\{\frac{D}{4}\left(1 - \frac{\cos\frac{\Theta}{2}\sin\frac{\Theta}{2}}{\Theta/2}\right)\right\}^{2/3} \tag{5.36}$$

In order to simulate the flow and depth variations in the pipes, the latter two equations are applied at successive points for successive time intervals.

It will be observed that it should never be necessary to consider surcharged conditions in a design. If pipes are designed to run just full at their design capacity, then they will run part full for any other storm duration. The higher up the leg a pipe length is, the shorter will be the concentration time, or time to flow equilibrium. The design storm duration will equal the concentration time of the drains down to the pipe in question. Any subsequent pipes will have larger concentration times and consequently a lower storm intensity.



Fig. 5.12 Cross section through part-full pipe.

Pipes higher up will therefore run part-full when a lower pipe is at its design capacity and running full.

The preceding scheme was employed in a program for analyzing the flow in each pipe in a drainage network the plan which is specified by the designer. The engineer must pre-select the layout, sub-division of catchment, position of inlets and grades. The grades will in general conform to the slope of the ground.

APPLICATION IN DESIGN

Few existing storm drain design methods allow for the increase in flow in the drain until equilibrium is reached. Nor is there often allowance for the fact that upper drains are designed for a more intense storm than lower drains. The upper drains have small concentration times and consequently the design storm duration is small. Lower drains will be designed for longer duration storms. Consequently the upper drains may not flow full when the lower drains are at their design capacity.

It is in fact necessary to simulate the flow overland and in each upper drain in order to size any particular lower drain. Such analysis can only be done practically by digital computer using numerical solutions of the flow equations. Many calculations are necessary for complex networks. A limitation on the maximum time interval for numerical stability implies many iterations until equilibrium flow conditions are reached for each pipe design. In addition, a number of different storm durations must be investigated for each pipe. A simple and efficient iterative procedure was therefore sought in order to minimize computer time. The kinematic form of the flow equations was employed to ensure this. The emphasis throughout the program is simplicity of data input and minimization of computational effort. Obviously some accuracy is sacrificed hereby, but the overriding assumption of precipitation pattern is probably more important. Sensitivity analysis and refinement can, if justified, be done with more sophisticated analytical models.

The simulation proceeds for successive pipes, the diameters of which are known. The same analytical procedure may be employed for design, that is the selection of pipe diameters. Starting at the top end of a drainage system, one sizes successively lower pipes. Thus each pipe upstream of the one to be designed, is defined. It is necessary to investigate storms of different duration and intensity of flow to determine the design storm resulting in maximum flow for the next pipe.

It is assumed that the design storm recurrence interval is pre-selected. The intensity-duration relationship is assumed to be of the form

$$i_e = \frac{a}{b + t_d} \qquad (5.37)$$

By selecting storms of varying duration t_d, and simulating the flow buildup down the drains, the program can select a storm which will result in the maximum peak flow from the lower end of the system. That discharge is the one to use for sizing the subsequent pipe. Thus the program proceeds from pipe to pipe until the entire network is designed. It should be noted that the network layout and pipe grades are pre-selected.

The sophistication of gradient optimization by dynamic programming (Merrit and Bogan, 1973; Dajani and Hasit, 1974) would add considerably to the computational cost. Other optimization techniques (Argaman et al, 1973; Yen and Sevuk, 1974) also add to the computations and omit the factor of decreasing storm intensity for lower sewers.

The algorithm was employed to design the drain size for the layout depicted in Fig. 5.13 . Input data and output are appended.



Fig. 5.13 Layout plan of drainage network sized in example.

COMPUTER PROGRAM FOR SIZING DRAIN PIPES IN NETWORKS

Assumptions in computer program

Pipes are assumed to flow initially at a depth corresponding to a subtended angle of 0.2π at the centre. The corresponding flow is very low, but this assumption avoids an anomaly for the case of zero depth when the numerical form of the equations is unrealistic.

Inflow from subcatchments is assumed to occur along the full length of the respective pipe, i.e. subcatchment breadth is assumed to be equal to pipe length. This affects overland flow time to some extent. If necessary (if flow is sensitive to storm duration) the subcatchment friction factor could be adjusted to give the correct overland flow time.

The computer program, written in FORTRAN for use in conversational mode on a terminal connected to an IBM 370 machine, is appended. The input format is described below. Data are read in free format and can be input on a terminal as the program stands.

First line of data:
M, A, B, E, IN, IR, II, G.
Second and subsequent lines of data (one line for each length of pipe):
X(I), S(I), Z(I), C(I), SO(I), EO(I), IB(I).

The input symbols are explained in the appended list and elaborated below:

M - The number of pipes: the number of pipes should be minimized for computational cost minimization. For computational accuracy the pipes should be divided into lengths of the same order of magnitude. It is convenient to make the pipe lengths equal to the distance between inlets. Inlets between 10 and 200 m apart are normally sufficient for computational accuracy. There should be at least two pipes in the system.

A,- Precipitation rate i is calculated from an equation of the form
B $i = A/(B + t_d)$ where t_d is the storm duration and B is a regional constant (both in seconds). A is a function of storm return period and catchment location and its units are in m if SI units are used, and ft if ft-lb-sec units are used.

E - Pipe roughness. This is analogous to the Nikuradse roughness and E is measured in m or ft. It is assumed in the program that all pipes have the same roughness. A conservative figure of at least 0.001 m (0.003 ft) is suggested to account for surface deterioration with time due to erosion, corrosion or deposits.

IN,-For each pipe sizing computation various storm durations are in-
IR vestigated, ranging from IUI up to IU2 in steps of IR (all in seconds). The smallest storm duration IU1 is set equal to the overland flow time for an upper pipe or the previous pipe design storm duration for subsequent pipes down a leg. The number of storm durations investigated is specified by IN and the increment in trial storm duration is specified by IR. Thus IU2 = IU1+IN*II. The accuracy of the computations is affected by the number of trial storm durations. A value of IN between 3 and 10 is usually satisfactory. The upper limit can be estimated beforehand from experience or by trial (if all design storm durations turn out to be less than the IU2 specified then the IN selected is satisfactory).

II -The computational time and cost is affected by the time increment of computations II (seconds). The maximum possible value is dependent on the numerical stability of the computations. A value equal to the minimum value of

$$\left[\frac{B}{C(I)A}\right]^{0.4} \quad \left[\frac{Z(I)E}{100\sqrt{S(I)G}\,X(I)}\right]^{0.6}$$

will normally be satisfactory (of the order of 60 to 300 seconds).

G - Gravitational acceleration (9.8 in SI units and 32.2 in ft - sec units).

The pipe data is next read in line by line for M pipes. As the program stands, 98 individual pipes are permitted, and any number of legs subject to the maximum number of pipes.

X(I) The pipe length in metres or ft, whichever units are used. An upper limit on individual pipes of 200 m is suggested for computational accuracy and a lower limit of 10 m for optimizing computer time.

S(I) The slope of that pipe in m per m or ft per ft.

Z(I) The surface area contributing runoff to the pipe in m^2 or ft^2.

C(I) The proportion of precipitation which runs off (analogous to the 'C' in the Rational formula).

SO(I) The overland slope of the contributing area, towards the inlet at the head of the pipe.

EO(I) The equivalent roughness of the overland area in m or ft depending on units employed.

IB(I) The number of the pipe which is a branch into the head of pipe I.

For no branch, put IB(I) = 0

For a header pipe at the top of a leg, put IB(I) = -1.

Only one branch pipe per inlet is permitted.

More must be accommodated by inserting short dummy pipes between.

The order in which pipes are tabulated should be obtained as follows:

After drawing out a plan of the catchment with each pipe, mark the longest leg possible starting from the outfall, then successively shorter legs on first the longest, then successively shorter pipes. Now number the pipes in the reverse order, starting at the top of the shortest leg of the shortest leg of the shortest leg, etc. Proceed down each leg with the numbering until a junction is reached. Never proceed past a branch which has not been tabulated previously. In this way all pipes leading into a pipe will have had their diameters calculated before the next lower pipe is designed.

Sample input

The data are in metres here and are taken from Fig. 5.13.

8	.075	1440	.001	3	300	60	9.8
100	.002	20000	.4	.005	.01	-1	
150	.004	20000	.4	.003	.01	-1	
200	.004	40000	.4	.003	.01	1	
100	.002	10000	.3	.005	.02	-1	
100	.004	40000	.4	.003	.01	-1	
200	.004	10000	.5	.005	.01	4	
200	.002	40000	.4	.002	.01	0	
100	.005	20000	.4	.003	.01	3	

Symbols in computer program

A Rainfall parameter in the equation: Precipitation rate = A/(B + IU). Metres or feet.
A1 Intermediate calculation variable (no significance).
A2 AT(I)/2
AT(I) Angle subtended at base of pipe by water surface. Radians.
B Time constant in the equation: Precipitation rate = A/(B+IU). Seconds.
C(I) Proportion of rain which runs off subcatchment I.
D(I) Diameter of pipe I. Metres or feet.
E Equivalent roughness of pipes. Metres or feet.
EO(I) Equivalent roughness of subcatchment surface I. Metres or feet.
G Gravitational acceleration 9.8 m/s² or 32.2 ft/sec².
I Pipe number.
IA(I) Feeder pipe 1 for pipe I.
IB(I) Branch pipe 2 for pipe I. If IB =-1, pipe I is a header.
IN Number of steps in rainfall duration.

II	Increment in time between iterations. Seconds.
IT	Iteration number.
IR	Increment in storm duration. Seconds.
IU	Storm duration. Seconds
IU1	Lower limit on storm duration. Seconds
IU2	Upper limit on storm duration. Seconds
IUI	Concentration time for overland flow. Seconds
J	Iteration number for overland flow calculation
M	Number of pipes
M1	M - 1
M2	Pipe number when iterating successive pipe diameters.
P(I)	Inflow to sewer from subcatchment. P(I) = Z(I)*C(I)*A/(B + IU). m^3/s or ft^3/sec.
PO	Excess rain rate from subcatchment. m/s or ft/sec.
PP1	Inflow rate from subcatchment. m^3/s or ft^3/s.
Q(I)	Flow rate in pipe. m^3/s or ft^3/s.
QO(I)	Overland flow per unit width of subcatchment. m^2/s or ft^2/s.
QOV	Intermediate calculation variable.
QP(I)	Flow rate in pipe for previous time interval. m^3/s or ft^3/s.
QQ(I)	Design flow in pipe. m^3/s or ft^3/s.
Q1,Q2	Intermediate calculation parameters.
S(I)	Slope of pipe. m/m or ft/ft.
SO(I)	Ground slope of subcatchment. m/m or ft/ft.
UU(I)	Design storm duration. Seconds.
X(I)	Length of pipe. m or ft.
Z(I)	Area of subcatchment. m^2 or ft^2.

REFERENCES

Argaman, Y., Shamir, U. and Spivak, E., 1973. Design of optimal sewerage systems. Proc. ASCE, (99), EE5. Oct., p 703-716.

Constandinides, C. A., 1981. Numerical techniques for a 2-dimensional kinematic overland flow model, Water S.A., Pretoria.

Dajani, J.S. and Hasit, Y., 1974. Capital cost minimization of drainage networks. Proc. ASCE, (100), EE2, April. p325-337.

Merritt, L.B., add Bogan, R.H., 1973. Computer based optimal design of sewer systems. Proc. ASCE, (99), EE1, Feb. p35-53.

Overton, D.E. and Meadows, M.E., 1976. Stormwater modelling. Academic Press, N.Y. 358pp.

Stephenson, D. 1980. Direct design algorithm for storm drain networks. Proc. Int. Conf. Urban Storm Drainage, Univ. Kentucky, Lexington.

Watkins, L.H., 1972. The design of urban sewer systems. Tech. Paper No. 55, Road Research Lab., London.

Wooding, R.A., 1965. A hydraulic model for the catchment stream problem. J. Hydrol. (a) Kinematic wave theory, (3), 1965, p254-257; (b) Numerical solutions, (3), 1965, p268-282; (c) Comparison with runoff observations (4), 1966, p21-37.

Woolhiser, D.A. and Liggett, J.A., 1967. Unsteady one-dimensional flow over a plane - the rising hydrograph. Water Resources Res., 3 (3), p753-771

Yen, B.C. and Sevuk, A.S., 1975. Design of storm sewer networks. Proc. ASCE, 101, EE4, Aug. p535-553.

Computer program for sizing storm drains in a network

```
      DIMENSION P(99),Q(99),S(99),X(99),D(99),AT(99),Z(99),QQ(99),UU(99),C(99)
      DIMENSION EO(99),SO(99),QO(99),IB(99),IA(99),QP(99)
      READ(9,5)M,A,B,E,IN,IR,II,G
5     FORMAT(8Y)
10    FORMAT(7Y)
      DO 15 I=1,M
      READ(9,10)X(I),S(I),Z(I),C(I),SO(I),EO(I),IB(I)
      IF(IB(I).GT.1)GO TO 13
      IF(IB(I).LT.0)GO TO 12
      IB(I)=99
      GO TO 13
12    IB(I)=99
      IA(I)=99
      GO TO 15
13    IA(I)=I-1
15    CONTINUE
      C(99)=0.
      QP(99)=0.
      QP(I)=0.
      M1=M-1
      DO 25 I=1,M
      IF(IA(I).LT.99)GO TO 25
      PO=C(I)*A/(B+500)
      DO 20 J=1,10
20    PO=C(I)*A/(B+(Z(I)/X(I)*EO(I)**.167/7.7/(SO(I)*G)**.5)**.6/PO**.4)
      IUI=C(I)*A/PO-B
      UU(I)=IUI
      PP1=PO*Z(I)
      QQ(I)=PP1
      D(I)=(PP1*E**.167/7.7/(S(I)*G)**.5/3.141*4**1.667)**.375
25    CONTINUE
      DO 300 M2=1,M1
      IU1=IUI
      IF(IA(M2+1).GE.99)GO TO 300
      QQ(M2+1)=0.
      IU2=IU1+IN*IR
      DO 200 IU=IU1,IU2,IR
      DO 30 I=1,M
      BIU=B+IU
      QO(I)=Z(I)/10./X(I)*C(I)*A/(BIU)
      AT(I)=.2*3.141
30    Q(I)=0.
      DO 120 IT=1,IU,II
      M3=M2+1
      DO 32 I=1,M3
      QOV=(7.7*(SO(I)*G)**.5/EO(I)**.167)**.6*1.667*II*QO(I)**.4
      QO(I)=QO(I)+QOV*(C(I)*A/(B+IU)-QO(I)/Z(I)*X(I))
32    P(I)=QO(I)*X(I)
      DO 100 I=1,M2
35    IF(IT.LE.1)GO TO 50
40    A1=(P(I)-Q(I)+QP(IA(I))+QP(IB(I)))/X(I)*II*8./D(I)/D(I)
45    AT(I)=A1/(1-(COS(AT(I)/2.))**2.+(SIN(AT(I)/2.))**2.)+AT(I)
50    A2=AT(I)/2.
      Q1=7.7*SQRT(G*S(I))/E**(1./6.)
      Q2=Q1*D(I)**2./4.*(A2-COS(A2)*SIN(A2))
100    Q(I)=Q2*(D(I)/4.*(1.-COS(A2)*SIN(A2)/A2))**(2./3.)
      DO 110 I=1,M2
110     QP(I)=Q(I)
120    CONTINUE
      Q(M2+1)=Q(IB(M2+1))+Q(M2)+P(M2+1)
      IF(Q(M2+1).LE.QQ(M2+1))GO TO 200
      QQ(M2+1)=Q(M2+1)
      UU(M2+1)=IU
200   CONTINUE
      D(M2+1)=(QQ(M2+1)*E**(1./6.)/7.7/SQRT(S(M2+1)*G)/3.141*4.**(5./3.))**(3./8.)
      IF(IA(M2+1).LT.99)GO TO 290
      IU1=IUI
      GO TO 300
290   IU1=UU(M2+1)
300   CONTINUE
      WRITE(5,350)
350    FORMAT(' STORM SEWER DESIGN')
      WRITE(5,60)
60    FORMAT(' PIPE LENGTH  DIA  GRADE  DSFLO/S  STORM S   AREA')
      DO 400 I=1,M
400   WRITE(5,70)I,X(I),D(I),S(I),QQ(I),UU(I),Z(I)
70    FORMAT(I6,F7.0,F6.3,F6.4,F9.3,F8.0,F9.0,F9.0)
      WRITE(5,80)M,A,B,E,IU2,IR,II
80    FORMAT(' DATA'I6,F5.3,F5.0,F5.4,3F6.0)
      STOP
      END
```

COMPUTER OUTPUT FOR SAMPLE RUN

```
STORM SEWER DESIGN
PIPE LENGTH  DIA  GRADE  DSFLO/S  STORM S   AREA
     1   100.  .576 .0020     .244   1014.   20000.
     2   150.  .514 .0040     .255    911.   20000.
     3   200.  .643 .0040     .462   2068.   40000.
     4   100.  .415 .0020     .102    772.   10000.
     5   100.  .574 .0040     .342   2068.   40000.
     6   200.  .619 .0040     .417   2068.   10000.
     7   200.  .853 .0020     .696   2068.   40000.
     8   100.  .905 .0050    1.287   2068.   20000.
DATA      8 .0751440..0010  2963    300     60
```